

Лабораторная работа №2

Движение РТР (Перемещения в прямоугольной системе координат)

Контроллер робота непосредственно управляет углом поворота звеньев робота. Однако, программирование в углах поворота для человека крайне неудобно - более привычна прямоугольная (декартова) система координат. Для этой цели контроллеры всех роботов в автоматическом режиме поддерживают преобразование координат (рис.1).




Рис.1

Пример программы для перемещения РТР в декартовой системе координат:

```
DEF LB_2 ()
```

```
$BASE = $WORLD ; установка системы координат рабочего пространства
$TOOL = $NULLFRAME ; установка системы координат инструмента
```

```
$VEL_AXIS[X]=100 ; установка скоростей
$VEL_AXIS[2]=100
$VEL_AXIS[3]=100
$VEL_AXIS[4]=100
$VEL_AXIS[5]=100
$VEL_AXIS[6]=100
```

```
$ACC_AXIS[1]=100 ; установка ускорений
$ACC_AXIS[2]=100
$ACC_AXIS[3]=100
$ACC_AXIS[4]=100
$ACC_AXIS[5]=100
$ACC_AXIS[6]=100
```

```
PTP XHOME
```

```
PTP {POS:X 1025,Y 0,Z 1480,A 0,B 90,C 0,S 'B 010',T 'B 000010'}
```

```
...
```

```
PTP XHOME
```

```
END
```




Рис. Заводская конфигурация СК инструмента NULLTOOL

Пример программы с заданием параметров движения путем вызова подпрограммы BAS:

```
DEF LB_2 ( )
```

```
;----- Declaration section (Секция объявления переменных) -----
```

```
EXT BAS (BAS_COMMAND :IN,REAL :IN ); объявление наличия внешней подпрограммы BAS
```

```
DECL AXIS HOME ;объявление переменной HOME типа AXIS
```

```
DECL FRAME BASE1 ;объявление переменной BASE1 типа FRAME
```

```
;----- Initialization (Секция инициализации переменных) -----
```

```
BAS (#INITMOV,0 ) ;вызов внешней подпрограммы BAS (см. файл bas.src) для инициализации скорости, ускорения, переменных $BASE, $TOOL и т.д.
```

```
HOME= {AXIS: A1 0,A2 -90,A3 90,A4 0,A5 0,A6 0}; присвоение значения переменной HOME
```

```
BASE1= {FRAME: X 300,Y -100,Z 0,A 90,B 0,C 0}; присвоение значения переменной BASE1
```

```
;----- Main section (Основная секция) -----
```

```
PTP HOME ; перемещение в исходную позицию
```

```
PTP {POS: X 540,Y 630,Z 1500,A 0,B 90,C 0,S 2,T 35}; перемещение в СК $BASE
```

```
PTP BASE1: {POS: X 540,Y 630,Z 1500,A 0,B 90,C 0,S 2,T 35}; перемещение относительно СК BASE1
```

```
PTP HOME
```

```
END
```

Вспомогательные системные подпрограммы расположены в системных файлах и уже готовы к использованию. Далее приведен текст лишь части файлов. Описание системных переменных см. в соответствующей документации.

Файл **bas.src** (см. полную исходную версию файла)

```
DEF BAS (COMMAND :IN, REAL_PAR :IN ); объявление подпрограммы и ее аргументов
```

```
...
```

```
DECL BAS_COMMAND COMMAND; объявление переменной COMMAND типа BAS_COMMAND (см. файл $config.dat)
```

```
REAL REAL_PAR; объявление переменной REAL_PAR вещественного типа
```

```
SWITCH COMMAND; оператор выбора
```

```
CASE #INITMOV
```

```
INITMOV ( ); вызов подпрограммы INITMOV
```

```
...
```

```
ENDSWITCH
```

```

END
...
DEF INITMOV ( ); объявление подпрограммы INITMOV
...
; PTP-Parameters (Параметры PTP-движений):
ACC_PTP (100.0 ) ; вызов подпрограммы задания ускорений ACC_PTP с передачей значения
VEL_PTP (100.0 ) ; вызов подпрограммы задания скорости VEL_PTP с передачей значения
...
END

```

Описание подпрограммы `ACC_PTP()` и `VEL_PTP()` находится в этом же файле **bas.src**

```

DEF ACC_PTP (ACC_FACTOR :IN ); подпрограмма для задания ускорений
REAL ACC_FACTOR ; объявление переменной ACC_FACTOR вещественного типа (проценты)
REAL ACC ; объявление переменной ACC вещественного типа
INT I ; объявление переменной-счетчика I для использования в циклах
...
IF $ADAP_ACC == #NONE THEN; проверка режима ускорений (возможны: #NONE, #STEP1, #STEP2)
    IF ACC_FACTOR > 100.0 THEN
        ACC=DEF_ACC_PTP ; присвоение переменной ACC значения по умолчанию, если задано >100
    ELSE
        ACC=DEF_ACC_PTP*ACC_FACTOR/100.0
    ENDIF
ELSE
    ACC=ACC_FACTOR
ENDIF

FOR I=1 TO $NUM_AX; задание ускорений для базовых осей робота
    $ACC_AXIS[I]=ACC
ENDFOR

FOR I=1 TO $EX_AX_NUM; задание ускорений для дополнительных осей робота
    $ACC_EXTAX[I]=ACC
ENDFOR
END

DEF VEL_PTP (VEL_FACTOR :IN ); подпрограмма для задания скоростей
REAL VEL_FACTOR ; в процентах [%]
REAL VEL
INT I

IF VEL_FACTOR > 100.0 THEN
    VEL=DEF_VEL_PTP
ELSE
    VEL=DEF_VEL_PTP*VEL_FACTOR/100.0
ENDIF

FOR I=1 TO $NUM_AX; задание скоростей для базовых осей робота
    $VEL_AXIS[I]=VEL
ENDFOR

FOR I=1 TO $EX_AX_NUM; задание скоростей для дополнительных осей робота
    $VEL_EXTAX[I]=VEL
ENDFOR
END

```

Файл \$config.dat

; PTP - movements (движения - PTP)

;-----

INT DEF_VEL_PTP=100; задание скоростей по умолчанию

INT DEF_ACC_PTP=50; задание ускорений по умолчанию

...

NUM BAS_COMMAND INITMOV, ACC_CP, ACC_PTP, VEL_CP, VEL_PTP, ACC_GLUE, TOOL,
BASE, EX_BASE, PTP_DAT, CP_DAT, OUT_SYNC, OUT_ASYNC, GROUP, FRAMES,
PTP_PARAMS, CP_PARAMS; объявление переменной BAS_COMMAND типа перечисление с
заданием возможных значений

...